mdepriv
Synthetic indicators of multiple deprivation

Marìa Noel Pi Alperin
CEPS/INSTEAD, Luxembourg
marianoel.pialperin@ceps.lu

Philippe Van Kerm†
CEPS/INSTEAD, Luxembourg‡
philippe.vankerm@ceps.lu

February 2009 (last revised March 2014)

Abstract This note describes `mdepriv`, a user-written Stata command for computing basic synthetic scores of multiple deprivation. The synthetic scores are weighted sums of individual unidimensional items taking values on the [0,1] range. Several alternative weighting rules are available.

Keywords `mdepriv` ; Stata ; multidimensional deprivation ; poverty

1 Introduction

We describe `mdepriv`, a Stata command to calculate basic synthetic scores of multiple deprivation from unidimensional indicators of deprivation. The command is available online for installation in net-aware Stata.[1] At the command prompt, type

```
ssc install mdepriv
```

2 Synthetic scores of multiple deprivation

`mdepriv` creates synthetic scores of deprivation that are linear functions of uni-dimensional deprivation items measured on the [0,1] range. Let $x_{ij} \in [0, 1]$ denote the value of a particular deprivation item $j \in \{1, \ldots, M\}$ for an observation $i \in \{1, \ldots, N\}$. The synthetic scores calculated by `mdepriv` are weighted sums of the M deprivation items:

$$s_i = \sum_{j=1}^{M} w_j x_{ij}.$$

Weights w_j determine the relative contribution of each item j to the aggregate index. This approach to constructing synthetic indices of deprivation is the simplest aggregative strategy; see, e.g., Deutsch & Silber (2005), Fusco (2005), Brandolini (2008) for reviews of this and other, more sophisticated approaches.

A number of different rules have been commonly used by practitioners to determine w_j. `mdepriv` implements four main weighting schemes. The first and simplest rule is 'equal weighting'

$$w_j^{eq} = \frac{1}{M}.$$

---

‡Centre d'Etudes de Populations, de Pauvreté et de Politiques Socio-Economiques/International Networks for Studies in Technology, Environment, Alternatives, Development. 3 Avenue de le fonte, L-4364 Esch/Alzette, Luxembourg. http://www.ceps.lu

[1] The latest version of the `mdepriv` package is 2.0.0 (of 2014-03-14). Stata 9.2 or later is required.

where s_i is simply the average value of x_{ij}. This is also known as the 'counting approach' and is underlying, for example, official EU statistics on material deprivation (Guio *et al.*, 2012).[2] The other rules implemented are 'frequency-based' where item weights are function of their sample mean.[3]

Desai & Shah (1988) used

$$\omega_j^{ds} \propto (1 - \bar{x}_j)$$

where $\bar{x}_j$ is the sample mean of item x_{ij}. This 'prevalence weighting' scheme is used for example in the UK 'Households Below Average Income' statistics on children and pensioners material deprivation (McKay, 2010, Department for Work and Pensions, 2013). Cerioli & Zani (1990) suggested a similar scheme but one giving stronger weight to relatively rare items as

$$\omega_j^{cz} \propto \log\left(\frac{1}{\bar{x}_j}\right).$$

Betti & Verma (1998) (and subsequently Betti *et al.*, 2008, Pi Alperin, 2007, 2008) adopted a more sophisticated double-weighting rule sensitive to both the relative frequency of items and the correlation among items. The correlation is taken into account so that two perfectly correlated items 'count as one' and only two uncorrelated items fully 'count as two'. To achieve this, Betti & Verma (1998) and Betti *et al.* (2008) defined item weights as the product of two components

$$\omega_j^{bv} \propto \left(\omega_j^a \times \omega_j^b\right)$$

with ω_j^a being the coefficient of variation of x_{ij} acting similarly to the frequency-based weights described above,[4]

$$\omega_j^a = \frac{\left(\sum_{i=1}^N (x_{ij} - \bar{x}_j)^2\right)^{1/2}}{\bar{x}_j N^{1/2}}$$

and

$$\omega_j^b = \left(1 + \sum_{m=1}^M \rho_{jm} I(\rho_{jm} < \rho_H)\right)^{-1} \left(\sum_{m=1}^M \rho_{jm} I(\rho_{jm} \geqslant \rho_H)\right)^{-1}$$

where ρ_{jm} is the correlation between items j and m and $I(\cdot)$ is an indicator function evaluating to 1 if the expression in brackets is true and 0 otherwise. ρ_H is a pre-determined cut-off correlation level.[5] ω_j^b is the inverse of a measure of 'average correlation' of item j with all the other items. The larger is the average correlation with item j, the lower is the resulting weight for item j.

In all cases, normalization to unity is achieved by setting

$$w_j = \frac{\omega_j}{\sum_{m=1}^M \omega_m}.$$

[2]In general, weights are normalized to sum to unity. In some circumstances however, weights can be taken to sum to M; strictly speaking, in a 'counting approach', $w_j = 1$.

[3]A key family of weighting rules not considered here involves estimating the weights from first-stage multivariate models (e.g., factor analysis, or principal components analysis). The first stage models also allow grouping of items into 'dimensions' (see *supra*). See, e.g., Callan *et al.* (1993), Whelan *et al.* (2001).

[4]In the case of dichotomous items coded 0/1, the coefficient of variation gives a weight function of the form: $\omega_j^a = \sqrt{\frac{1}{\bar{x}_j} - 1}$. This effectively gives higher weight to relatively infrequent deprivation items, as in the Cerioli-Zani weighting scheme.

[5]ρ_H separates high and low correlations. Betti & Verma (1998) suggest setting ρ_H as to divide the ordered set of correlations at the point of the largest gap.

3 A decomposable index of aggregate deprivation

Once synthetic scores s_i for all sample observations are estimated, a simple measure of aggregate deprivation is the sample mean:

$$S = \frac{1}{N}\sum_{i=1}^{N} s_i.$$

Because s_i is a linear function of individual-level item deprivation measures x_{ij}, it is straightforward to factorize S into contributions from each of the items (see Pi Alperin, 2007, 2008):

$$S = \sum_{j=1}^{M} S_j$$

where

$$S_j = w_j \bar{x}_j.$$

The relative contribution of item j to the aggregation deprivation measure S, namely S_j/S, is the product of its weight and the average value of item j in the sample divided by the aggregate index.

4 Hierarchical structures: Items vs. dimensions

In many applications, deprivation items can be grouped in a smaller number of K dimensions.

This has implications on the normalization of weights. With hierarchical structures, item weights are typically normalized to sum to unity within each dimension so that each dimension has equal weight, irrespectively of the number of items it contains:

$$w_j = \frac{\omega_j}{\sum_{m\in\Omega_j}\omega_m}.$$

where Ω_j is the set of indices of items belonging to the same dimension as item j. With this normalization, the contribution of each dimension does not depend on the number of items in the dimension. The sum of all weights is equal to K, the number of dimensions. Alternatively, one can further divide by K to normalize the overall sum of weights to sum to unity,

$$w_j = \frac{1}{K}\frac{\omega_j}{\sum_{m\in\Omega_j}\omega_m}.$$

Grouping items in dimensions also has implications for the Betti & Verma (1998) weighting scheme. In computing the ω_j^b factor, only correlations between items of the same dimension are taken into account. Correlation of items of different dimensions does not affect item weights.

5 The `mdepriv` command

The user-written command `mdepriv` generates a variable containing observation-level synthetic indicators s_i from a series of variables measuring individual deprivation items, optionally grouped in dimensions. It also reports the aggregate measure S and the contribution of each of the separate items and dimensions, if relevant.

5.1 Syntax

`mdepriv` [( ] *varlist* [ )] [(*varlist*) ...(*varlist*)] [*if*] [*in*] [*weight*] [, *options*]

`aweight` and `fweight` are allowed.

Each element in *varlist* is a variable containing the deprivation value for a particular item. Variables can be grouped in dimensions by putting blocks of variables in brackets.

5.2 Options

`generate(`*newvarlist* [`, replace`]`)` specifies the name of a new variable to store the synthetic deprivation score.

`method(`*string*`)` selects the weighting scheme. Available choices are `equal` for equal weighting of items, `ds` for the Desai-Shah weighting, `cz` for the Cerioli-Zani weighting, and `bv` for the Betti-Verma weighting. The latter choice allows a `rhoh(`#`)` sub-option to choose ρ_H and one of sub-options `pearson`, `tetrachoric`, `polychoric` or `mixed` (the default) to select a correlation type.[6] Finally, `vec(`*namelist*`)` can be used to pass user-defined weights: *namelist* is a list of K (column) matrix names containing the weights for the items in each of the K dimensions. The number of matrix names must match the number of dimensions and weights in each matrix are applied to items in the same order as items are passed in *varlist*. Default is `method(cz)`.

`wa(`*string*`)` and `wb(`*string*`)` provide alternative, more flexible ways to select the weighting schemes. Weights are computed as the product of two terms as in the Betti-Verma scheme. `wa(`*string*`)` selects the form of the first factor and is one of `ds`, `cz`, `bv` or `equal`. `wb(`*string*`)` selects the form of the second factor and is one of `pearson`, `tetrachoric`, `polychoric`, `mixed` or `diagonal` where the latter discards all correlations. Specify either `wa(`*string*`)` and `wb(`*string*`)` or `method(`*string*`)`, not both.

`force` allows calculations even if items are not limited to the $[0, 1]$ range.

`install` checks if required user-written packages `polychoric` and `ozsutils` are installed, and prompts for installation if needed.

5.3 Saved results

`mdepriv` is an r-class command and returns the following values:

Scalars

`r(N)`	number of observations	`r(sum_w)`	sum of weights
`r(ndim)`	number of dimensions	`r(aggregate)`	aggregate index S

Macros

`r(itemslist)`	list of items	`r(generate)`	created variable
`r(weightschemename)`	label of weighting scheme	`r(method)`	selected method
`r(wa)`	value of `wa`	`r(wb)`	value of `wb`

Matrices

`r(fullweights)`	item weights	`r(fullindices)`	item means
`r(fullcontris)`	item contributions	`r(fullshares)`	relative item contributions

Additionally, for each dimension $i \in 1, \ldots, K$, `mdepriv` returns

Scalars

`r(dim_i)`	number of items in dimension	`r(index_i)`	within-dimension index
`r(contri_i)`	dimension contribution	`r(share_i)`	dimension relative contribution

Matrices

`r(itemweights_i)`	item weights within dimension	`r(itemcorrelations_i)`	item correlation matrix

5.4 Dependencies on user-written packages

`mdepriv` requires one other user-written package.

The `polychoric` package by Stas Kolenikov available from `http://web.missouri.edu/~kolenikovs/stata` is required to compute the correlation matrix with non-continuous and non-dichotomous items

[6] Note that Pearson's correlation coefficient computed over pairs of dichotomous variables is equivalent to Cramer's V coefficient of association.

in the Betti-Verma weighting scheme (that is, with options `method(bv, polychoric)` or `method(bv, mixed)`). See Kolenikov & Angeles (2009).

The package can be installed easily with the `install` options.

6 Examples

We briefly illustrate usage of `mdepriv` with data from the PSELL-3/EU-SILC survey *Liewen zu Letzebuerg*. The examples presented here are purely illustrative of usage of the `mdepriv` command. Pi Alperin (2008) or Hildebrand *et al.* (2012) provide a detailed and more substantive analysis of these data.

Our dataset contains observations on ten variables coded 0/1, all picking up a particular deprivation item: whether person can afford a colour TV, a PC, or a car, whether can afford medical care or a visit to the dentist if needed, whether can afford to keep house warm, whether the accommodation has problems such as a leaky roof, rot, whether the person lives in an area with noise, pollution or crime problems.

```
. su coltv PC car  med dentist  warm housingpbm noise pollution crime
```

Variable	Obs	Mean	Std. Dev.	Min	Max
coltvca	29940	.0011022	.0331817	0	1
PCca	29940	.041149	.198638	0	1
carca	29940	.0188043	.1358355	0	1
medca	29940	.004342	.0657519	0	1
dentistca	29940	.0076486	.0871228	0	1
warmca	29940	.0117234	.1076401	0	1
housingpbm	29940	.1755845	.3804726	0	1
noise	29940	.2346025	.4237572	0	1
pollution	29940	.1802605	.3844107	0	1
crime	29940	.1532064	.3601924	0	1

`mdepriv` can be used to aggregate these ten indicators into a single synthetic measure of deprivation using the weighted sums described *infra*. In a first example, we do not group these items in dimensions and use the default Cerioli-Zani weighting scheme. The second example assumes a dimensional structure; this has implications on the weight assigned to the different items.

```
. mdepriv coltv PC car med dentist warm housingpbm noise pollution crime
Cerioli & Zani (1990) weighting scheme
Aggregate deprivation level:   0.0442
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
coltvca	0.0011	0.1918	0.0002	0.0048
PCca	0.0411	0.0898	0.0037	0.0837
carca	0.0188	0.1119	0.0021	0.0476
medca	0.0043	0.1532	0.0007	0.0151
dentistca	0.0076	0.1372	0.0010	0.0238
warmca	0.0117	0.1252	0.0015	0.0332
housingpbm	0.1756	0.0490	0.0086	0.1948
noise	0.2346	0.0408	0.0096	0.2169
pollution	0.1803	0.0482	0.0087	0.1969
crime	0.1532	0.0528	0.0081	0.1833
Total		1.0000	0.0442	1.0000

```
. mdepriv (coltv PC car) (med dentist) (warm housingpbm) (noise pollution crime
```

```
> ), generate(s)
Cerioli & Zani (1990) weighting scheme
Aggregate deprivation level:   0.0662
Deprivation level, weight and contribution to total, by dimension
```

	Index	Weight	Contri	Share
Dimension 1	0.0153	0.2500	0.0038	0.0577
Dimension 2	0.0059	0.2500	0.0015	0.0223
Dimension 3	0.0578	0.2500	0.0145	0.2183
Dimension 4	0.1858	0.2500	0.0465	0.7017
Total		1.0000	0.0662	1.0000

```
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
coltvca	0.0011	0.1218	0.0001	0.0020
PCca	0.0411	0.0571	0.0023	0.0355
carca	0.0188	0.0711	0.0013	0.0202
medca	0.0043	0.1319	0.0006	0.0086
dentistca	0.0076	0.1181	0.0009	0.0136
warmca	0.0117	0.1797	0.0021	0.0318
housingpbm	0.1756	0.0703	0.0123	0.1865
noise	0.2346	0.0719	0.0169	0.2549
pollution	0.1803	0.0850	0.0153	0.2314
crime	0.1532	0.0931	0.0143	0.2154
Total		1.0000	0.0662	1.0000

```
. summarize s
```

Variable	Obs	Mean	Std. Dev.	Min	Max
s	29940	.0662034	.0885504	0	.75

The `method()` option allows switching weighting scheme. Five possibilities are illustrated here, including using sub-options to `method(bv)` and passing user-defined weights. As the results show, modifying the weighting scheme can have a major impact on the aggregate value of the deprivation index.

```
. mdepriv (coltv PC car) (med dentist) (warm housingpbm) (noise pollution crime
> ), method(ds) generate(s, replace)
Desai & Shah (1988) weighting scheme
Aggregate deprivation level:   0.0751
Deprivation level, weight and contribution to total, by dimension
```

	Index	Weight	Contri	Share
Dimension 1	0.0201	0.2500	0.0050	0.0669
Dimension 2	0.0060	0.2500	0.0015	0.0200
Dimension 3	0.0862	0.2500	0.0216	0.2872
Dimension 4	0.1879	0.2500	0.0470	0.6259
Total		1.0000	0.0751	1.0000

```
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
coltvca	0.0011	0.0850	0.0001	0.0012
PCca	0.0411	0.0816	0.0034	0.0447
carca	0.0188	0.0835	0.0016	0.0209
medca	0.0043	0.1252	0.0005	0.0072
dentistca	0.0076	0.1248	0.0010	0.0127
warmca	0.0117	0.1363	0.0016	0.0213
housingpbm	0.1756	0.1137	0.0200	0.2660
noise	0.2346	0.0787	0.0185	0.2459
pollution	0.1803	0.0843	0.0152	0.2024
crime	0.1532	0.0870	0.0133	0.1777
Total		1.0000	0.0751	1.0000

```
. mdepriv (coltv PC car) (med dentist) (warm housingpbm) (noise pollution crime
> ), method(bv)
```

Betti & Verma (1998) weighting scheme

Aggregate deprivation level: 0.0600

Deprivation level, weight and contribution to total, by dimension

	Index	Weight	Contri	Share
Dimension 1	0.0084	0.2500	0.0021	0.0351
Dimension 2	0.0058	0.2500	0.0014	0.0240
Dimension 3	0.0430	0.2500	0.0108	0.1792
Dimension 4	0.1828	0.2500	0.0457	0.7616
Total		1.0000	0.0600	1.0000

Deprivation level, weight and contribution to total, by item

	Index	Weight	Contri	Share
coltvca	0.0011	0.1815	0.0002	0.0033
PCca	0.0411	0.0278	0.0011	0.0190
carca	0.0188	0.0408	0.0008	0.0128
medca	0.0043	0.1427	0.0006	0.0103
dentistca	0.0076	0.1073	0.0008	0.0137
warmca	0.0117	0.2023	0.0024	0.0395
housingpbm	0.1756	0.0477	0.0084	0.1397
noise	0.2346	0.0658	0.0154	0.2573
pollution	0.1803	0.0751	0.0135	0.2256
crime	0.1532	0.1091	0.0167	0.2787
Total		1.0000	0.0600	1.0000

```
. mdepriv (coltv PC car) (med dentist) (warm housingpbm) (noise pollution crime
> ), method(bv , pearson rhoh(0.5) )
```

Betti & Verma (1998) weighting scheme

Aggregate deprivation level: 0.0602

Deprivation level, weight and contribution to total, by dimension

	Index	Weight	Contri	Share
Dimension 1	0.0078	0.2500	0.0020	0.0324
Dimension 2	0.0058	0.2500	0.0014	0.0239

	Index	Weight	Contri	Share
Dimension 3	0.0430	0.2500	0.0108	0.1786
Dimension 4	0.1842	0.2500	0.0461	0.7650
Total		1.0000	0.0602	1.0000

```
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
coltvca	0.0011	0.1873	0.0002	0.0034
PCca	0.0411	0.0254	0.0010	0.0174
carca	0.0188	0.0373	0.0007	0.0116
medca	0.0043	0.1427	0.0006	0.0103
dentistca	0.0076	0.1073	0.0008	0.0136
warmca	0.0117	0.2023	0.0024	0.0394
housingpbm	0.1756	0.0477	0.0084	0.1392
noise	0.2346	0.0687	0.0161	0.2677
pollution	0.1803	0.0799	0.0144	0.2392
crime	0.1532	0.1014	0.0155	0.2581
Total		1.0000	0.0602	1.0000

```
. mdepriv (coltv PC car) (med dentist) (warm housingpbm) (noise pollution crime
> ), method(equal)

Equi-proportionate weighting scheme

Aggregate deprivation level:   0.0773

Deprivation level, weight and contribution to total, by dimension
```

	Index	Weight	Contri	Share
Dimension 1	0.0204	0.2500	0.0051	0.0658
Dimension 2	0.0060	0.2500	0.0015	0.0194
Dimension 3	0.0937	0.2500	0.0234	0.3027
Dimension 4	0.1894	0.2500	0.0473	0.6121
Total		1.0000	0.0773	1.0000

```
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
coltvca	0.0011	0.0833	0.0001	0.0012
PCca	0.0411	0.0833	0.0034	0.0443
carca	0.0188	0.0833	0.0016	0.0203
medca	0.0043	0.1250	0.0005	0.0070
dentistca	0.0076	0.1250	0.0010	0.0124
warmca	0.0117	0.1250	0.0015	0.0189
housingpbm	0.1756	0.1250	0.0219	0.2838
noise	0.2346	0.0833	0.0196	0.2528
pollution	0.1803	0.0833	0.0150	0.1942
crime	0.1532	0.0833	0.0128	0.1651
Total		1.0000	0.0773	1.0000

```
. mat def w1 = (.2 \ .7 \ .1)

. mat def w2 = (.5 \ .5)

. mat def w3 = (.9 \ .1)
```

```
. mat def w4 = (.33 \ .33 \ .34)
. mdepriv (coltv PC car) (med dentist) (warm housingpbm) (noise pollution crime
> ), method(vec(w1 w2 w3 w4))
User-defined weighting scheme
Aggregate deprivation level:   0.0635
Deprivation level, weight and contribution to total, by dimension
```

	Index	Weight	Contri	Share
Dimension 1	0.0309	0.2500	0.0077	0.1217
Dimension 2	0.0060	0.2500	0.0015	0.0236
Dimension 3	0.0281	0.2500	0.0070	0.1107
Dimension 4	0.1890	0.2500	0.0472	0.7441
Total		1.0000	0.0635	1.0000

```
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
coltvca	0.0011	0.0500	0.0001	0.0009
PCca	0.0411	0.1750	0.0072	0.1134
carca	0.0188	0.0250	0.0005	0.0074
medca	0.0043	0.1250	0.0005	0.0085
dentistca	0.0076	0.1250	0.0010	0.0151
warmca	0.0117	0.2250	0.0026	0.0415
housingpbm	0.1756	0.0250	0.0044	0.0691
noise	0.2346	0.0825	0.0194	0.3048
pollution	0.1803	0.0825	0.0149	0.2342
crime	0.1532	0.0850	0.0130	0.2051
Total		1.0000	0.0635	1.0000

A more flexible way to set the weights is to use options `wa()` and `wb()` as an alternative to `method()`. In the following example, we first replicate the `method(bv)` option by setting `wa(bv)` and `wb(mixed)`. We then combine the Cerioli-Zani frequency-based weight factor with a correlation-based weighting factor à la Betti-Verma.

```
. mdepriv (coltv PC car) (med dentist) (warm housingpbm) (noise pollution crime
> ), wa(bv) wb(mixed)
User-defined weighting scheme
Aggregate deprivation level:   0.0600
Deprivation level, weight and contribution to total, by dimension
```

	Index	Weight	Contri	Share
Dimension 1	0.0084	0.2500	0.0021	0.0351
Dimension 2	0.0058	0.2500	0.0014	0.0240
Dimension 3	0.0430	0.2500	0.0108	0.1792
Dimension 4	0.1828	0.2500	0.0457	0.7616
Total		1.0000	0.0600	1.0000

```
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
coltvca	0.0011	0.1815	0.0002	0.0033

PCca	0.0411	0.0278	0.0011	0.0190
carca	0.0188	0.0408	0.0008	0.0128
medca	0.0043	0.1427	0.0006	0.0103
dentistca	0.0076	0.1073	0.0008	0.0137
warmca	0.0117	0.2023	0.0024	0.0395
housingpbm	0.1756	0.0477	0.0084	0.1397
noise	0.2346	0.0658	0.0154	0.2573
pollution	0.1803	0.0751	0.0135	0.2256
crime	0.1532	0.1091	0.0167	0.2787
Total		1.0000	0.0600	1.0000

```
. mdepriv (coltv PC car) (med dentist) (warm housingpbm) (noise pollution crime
> ), wa(cz) wb(mixed)
User-defined weighting scheme
Aggregate deprivation level:   0.0654
Deprivation level, weight and contribution to total, by dimension
```

	Index	Weight	Contri	Share
Dimension 1	0.0149	0.2500	0.0037	0.0571
Dimension 2	0.0059	0.2500	0.0015	0.0226
Dimension 3	0.0578	0.2500	0.0145	0.2211
Dimension 4	0.1828	0.2500	0.0457	0.6992
Total		1.0000	0.0654	1.0000

```
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
coltvca	0.0011	0.1254	0.0001	0.0021
PCca	0.0411	0.0561	0.0023	0.0353
carca	0.0188	0.0685	0.0013	0.0197
medca	0.0043	0.1319	0.0006	0.0088
dentistca	0.0076	0.1181	0.0009	0.0138
warmca	0.0117	0.1797	0.0021	0.0322
housingpbm	0.1756	0.0703	0.0123	0.1889
noise	0.2346	0.0659	0.0155	0.2367
pollution	0.1803	0.0753	0.0136	0.2077
crime	0.1532	0.1087	0.0167	0.2549
Total		1.0000	0.0654	1.0000

The last example demonstrates the flexibility of the command. We want to use the Betti-Verma weighting scheme to construct the item weights by dimension. But instead of using an equal weighting of all dimensions to construct the synthetic index, we want to apply the Cerioli-Zani weighting to each dimension, so that dimensions are not equally weighted in the aggregate, but weighted according to the average deprivation in this dimension. This can be achieved by repeated calls to `mdepriv` for each dimension separately and then calling `mdepriv` on the synthetic indicators for each dimension. This also illustrates that the command need not be applied to 0/1 data, but to any data defined on the [0,1] range. We first replicate results with the equal weighting of dimensions and then show the Cerioli-Zani weighting of dimensions.

```
. loc i 0
```

```
. foreach block in "coltv PC car" "med dentist"  "warmca housingpbm"  "noise po
> llution crime" {
  2.   qui mdepriv `block´ , method(bv) generate(s`++i´)
  3. }
. mdepriv s1 s2 s3 s4 , method(equal)
Equi-proportionate weighting scheme
Aggregate deprivation level:   0.0600
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
s1	0.0084	0.2500	0.0021	0.0351
s2	0.0058	0.2500	0.0014	0.0240
s3	0.0430	0.2500	0.0108	0.1792
s4	0.1828	0.2500	0.0457	0.7616
Total		1.0000	0.0600	1.0000

```
. mdepriv s1 s2 s3 s4 , method(cz)
Cerioli & Zani (1990) weighting scheme
Aggregate deprivation level:   0.0349
Deprivation level, weight and contribution to total, by item
```

	Index	Weight	Contri	Share
s1	0.0084	0.3231	0.0027	0.0781
s2	0.0058	0.3489	0.0020	0.0576
s3	0.0430	0.2129	0.0092	0.2623
s4	0.1828	0.1150	0.0210	0.6020
Total		1.0000	0.0349	1.0000

References

Betti, G., Cheli, B., Lemmi, A. & Verma, V. K. (2008), 'The fuzzy set approach to the multidimensional poverty: The case of Italy in the 1990s', in N. Kakwani & J. Silber (eds.), *Quantitative Approaches to Multidimensional Poverty Measurement*, Palgrave Mac Millan, New York, 30–48.

Betti, G. & Verma, V. K. (1998), 'Measuring the degree of poverty in a dynamic and comparative context: a multi-dimensional approach using fuzzy set theory', Working Paper 22, Dipartimento di Metodi Quantitativi, Università di Siena.

Brandolini, A. (2008), 'On applying synthetic indices of multidimensional well-being: Health and income inequalities in selected EU countries', Temi di discussione 668, Banca d'Italia.

Callan, T., Nolan, B. & Whelan, C. T. (1993), 'Resources, deprivation and the measurement of poverty', *Journal of Social Policy*, **22**(22):141–172.

Cerioli, A. & Zani, S. (1990), 'A fuzzy approach to the measurement of poverty', in C. Dagum & M. Zenga (eds.), *Income and Wealth Distribution, Inequality and Poverty*, Springer Verlag, Berlin, 272–284.

Department for Work and Pensions (2013), *Households Below Average Income: An analysis of the income distribution 1994/95 2011/12*, Department for Work and Pensions, London, UK, `https://www.gov.uk/government/collections/households-below-average-income-hbai--2`.

Desai, M. & Shah, A. (1988), 'An econometric approach to the measurement of poverty', *Oxford Economic Papers*, **40**(3):505–522.

Deutsch, J. & Silber, J. (2005), 'Measuring multidimensional poverty: An empirical comparison of various approaches', *Review of Income and Wealth*, **51**(1):145–174.

Fusco, A. (2005), *La Contribution des Analyses Multidimensionnelles à la Compréhension et à la Mesure du Concept de Pauvreté : Application Empirique au Panel Communautaire des Ménages*, Ph.D. thesis, Université de Nice-Sophia Antipolis.

Guio, A.-C., Gordon, D. & Marlier, E. (2012), 'Measuring material deprivation in the EU: Indicators for the whole population and child-specific indicators', Eurostat Methodologies and Working Papers KS-RA-12-018-EN-N, European Commission, Publications Office of the European Union, Luxembourg.

Hildebrand, V., Pi Alperin, M. N. & Vna Kerm, P. (2012), 'Measuring and accounting for the deprivation gap of Portuguese immigrants in Luxembourg', CEPS/INSTEAD Working Paper Series 2012-33, CEPS/INSTEAD.

Kolenikov, S. & Angeles, G. (2009), 'Socioeconomic status measurement with discrete proxy variables: Is principal component analysis a reliable answer?', *Review of Income and Wealth*, **55**(1):126–165.

McKay, S. (2010), 'Using the new Family Resources Survey question block to measure material deprivation among pensioners', DWP Working Paper 89, Department for Work and Pensions, London, UK.

Pi Alperin, M. N. (2007), *Mesure de la décomposition multidimensionnelle d'un indice de pauvreté base sur la théorie des ensembles flous. Le cas de l'Argentine de 1974 à 2003*, Ph.D. thesis, LAMETA, Université de Montpellier 1.

— (2008), 'A comparison of multidimensional deprivation characteristics between natives and immigrants in Luxembourg', IRISS Working Paper 2008-14, CEPS/INSTEAD, Differdange, Luxembourg.

Whelan, C. T., Layte, R., Maître, B. & Nolan, B. (2001), 'Income, deprivation and economic strain. an analysis of the European Community Household Panel', *European Sociological Review*, **17**(4):357–371.

Citation, liability, conditions of use

The program should work as advertised, but it is freely offered 'as-is' to the research community and may contain bugs. Use at your own risk! Of course, we appreciate bug reports, as well as comments and suggestions (email `philippe.vankerm@ceps.lu`).

Please cite as:
Pi Alperin, M. N. and Van Kerm, P. (2009), '`mdepriv` – Synthetic indicators of multiple deprivation', v2.0 (revised March 2014), CEPS/INSTEAD, Esch/Alzette, Luxembourg.

Acknowledgements

This work is part of the MeDIM project (*Advances in the Measurement of Discrimination, Inequality and Mobility*) supported by the Luxembourg 'Fonds National de la Recherche' (contract FNR/06/15/08) and by core funding for CEPS/INSTEAD by the Ministry of Higher Education and Research of Luxembourg. Extensive testing, comments and bug reports by Alessio Fusco are gratefully acknowledged.